

# ODS RTFとREPORTプロシジャを用いた 一覧表の出力

今野 聡

イーピーエス株式会社 統計解析1部

How to create lists by using  
ODS RTF / REPORT procedure.

Satoshi Konno

Statistics Analysis Division 1, EPS Co., Ltd.



## 要旨:

SAS9.2で拡張されたODS RTFの機能とREPORTプロシジャを用いたセル形式のRTF一覧表作成プロセスを紹介致します。

キーワード: ODS RTF , REPORTプロシジャ



# 結論



# このようなRTF(Rich Text Format)を作成

## 16.2.7 被験者ごとの有害事象一覧

被験者 番号	投与群 (mg/kg)	年齢 (歳)	BMI <sup>(1)</sup>	有害事象			
				事象名(PT) (MedDRA/J Ver XX.X)	日数 <sup>(2)</sup>	因果関係	コメント
001	10	15	18.5	下腹部痛	9	関連なし	原因不明ではあるが、偶発症状と思われる
				悪心	11	関連あり	XXXXXXXXXXXX投与後に発現しているため、治験薬に関連あり
002	10	20	18.2	γ-グルタミルトラン スフェラーゼ増加	125	関連あるか もしれない	XXXXXXXXXXXX投与のための影響 と思われるが、治験薬による 影響も否定できないと考える
				倦怠感	126	関連なし	XXXXXXXXXXXX療法に伴う影響
				血中アルカリホスファ ターゼ増加	131	関連なし	抗生物質の使用に伴う影響
003	20	33	28.4	結膜出血	154	関連なし	偶発性と思われる

(1: BMI=体重(kg)/(身長(cm)/100)<sup>2</sup>

(2: 投与開始日を1日目とした日数



# 導入

# RTFとは

## RTFとは何か

- ・Rich Text Formatの略
- ・Microsoft社により策定されている文書フォーマット
- ・Wordとの親和性が高い
- ・2011年4月時点では、最新版はRTF1.9
- ・英語の仕様書がWEB上に公開されている
- ・日本語資料が少ない
- ・メモ帳などでソースを開いて変更することが可能
  - ↳ 編集可能な点をメインに置いた仕様

被験者番号...	FAS...
XXX-XXX...	○...

ソース変更

被験者番号...	FAS...
XXX-X XX...	○...

メモ帳などでRTFを開いたソース  
...¥keepn¥sb29¥sa29¥qc¥fs16¥...

メモ帳などでソースを直接変更  
...¥keepn**¥b**¥sb29¥sa29¥qc**¥fs32**¥cf1...



# 発表の想定環境

## SASの実行環境

SAS 9.2を想定

- ・SAS9.2ではODS RTFの機能が拡張されている
  - ・CALL DEFINEステートメントによる罫線の追加
  - ・STYLEによるunicodeの対応

## RTFの閲覧

Word 2003もしくはWord 2007を想定

- ・編集中的見た目が若干異なるが、印刷結果を含め大きな差はない

## 出力プロシジャ

REPORTプロシジャを使用


- ・文字列の出力が可能
- ・セル単位の出力ができるため、Excelへの転載が容易
- ・罫線を含めセルごとの書式のコントロールが可能
- ・列のグループ化など見出しの加工が容易



# 弊社での解析結果作成の現状

## 弊社の現状


- ・弊社では解析結果のほとんどはExcelへの出力
- ・Excelへの出力は罫線や書式などを含めDDEを使用
- ・Excelへ出力する上で技術的に大きな問題はない




RTF出力は業務での使用経験はあるものの  
まだまだ試行錯誤を行っている

## RTFの活用を 広めるには...

Excelの他、出力媒体の選択肢として  
もっと活用の幅を広げたい



印刷結果を見る限りではExcelもRTFも変わらない  
使い分けるメリットは？



メリット・デメリットの整理






# メリット1: 総括報告書(CSR)などへの挿入が容易


Excel

・ドキュメント形式に変換する加工が必要



RTF

・Wordと親和性が高いため、そのまま挿入可能



# メリット2: 編集画面と印刷結果が同一

Excel

- ・Excelでは編集画面と印刷結果が異なる場合がある
- ↳ 結果作成後に文字切れ、ページ切れの確認が必要  
(フォントやスタイルの事前の設定である程度の対応は可能)

コメント
2011/5/17 症状悪化のため XXXXXXXXXXXXXXXXXXXX (XXXXXXXXXXXXXXXX)施術、 2011/5/18 症状が回復傾向

印刷

コメント
2011/5/17 症状悪化のため XXXXXXXXXXXXXXXXXXXX (XXXXXXXXXXXXXXXX)施術、 2011/5/18 症状が回復傾向

文字切れ

RTF

- ・Word上では編集画面と印刷結果が同一になる
- ・REPORTプロシジャの出力はセル形式で自動的にセル内改行される
- ↳ 文字切れ、ページ切れがない

コメント
2011/5/17 症状悪化のため XXXXXXXXXXXXXXXXXXXX (XXXXXXXXXXXXXXXX)施術、 2011/5/18 症状が回復傾向

印刷

コメント
2011/5/17 症状悪化のため XXXXXXXXXXXXXXXXXXXX (XXXXXXXXXXXXXXXX)施術、 2011/5/18 症状が回復傾向

同一の出力

文字切れやページ切れがないのが当然と扱われるが、  
EXCELでの解析結果作成時には  
全ページの確認には少なからず工数を要している


# メリット3: フォント、書式の統一

Excel

- ・事前にフォントや書式、スタイルの設定が必要  
(事後の作業としてExcel4.0マクロやVBAなどである程度の統一は可能)

RTF

- ・ユーザースタイルを設定してフォントや書式の統一が可能
- ・表ごとのカスタマイズは各プログラム中で対応可能



プログラム α

検体番号	検体量 (ng/kg)	年齢 (歳)	BMI <sup>①</sup>	検査名 (PT)	日数 <sup>②</sup>	回収率 (%)	コメント
001	10	15	18.5	下腹部痛	9	測定なし	測定不能ではあるが、検体状況と見られる
				重心	11	測定あり	XXXXXXXXXX検体後に検出しているため、治療に継続あり
002	10	20	18.2	アークリミルトロン スコア増加	125	測定あり	XXXXXXXXXX検体の元の検体もしれないが、治療による影響も否定できないと考え
				倦怠感	126	測定なし	XXXXXXXXXX療法に伴う影響
				血中アルカリホスファターゼ増加	131	測定なし	血中物質の検出に伴う影響
003	20	33	28.4	経尿出血	154	測定なし	検出なしと見られる

(1)BMI=体重(kg)/身長(m)<sup>2</sup>  
(2)検体検出日を日数とした日数

プログラム α

検体番号	検体量 (ng/kg)	年齢 (歳)	BMI <sup>①</sup>	検査名 (PT)	日数 <sup>②</sup>	回収率 (%)	コメント
001	10	15	18.5	下腹部痛	9	測定なし	測定不能ではあるが、検体状況と見られる
				重心	11	測定あり	XXXXXXXXXX検体後に検出しているため、治療に継続あり
002	10	20	18.2	アークリミルトロン スコア増加	125	測定あり	XXXXXXXXXX検体の元の検体もしれないが、治療による影響も否定できないと考え
				倦怠感	126	測定なし	XXXXXXXXXX療法に伴う影響
				血中アルカリホスファターゼ増加	131	測定なし	血中物質の検出に伴う影響
003	20	33	28.4	経尿出血	154	測定なし	検出なしと見られる

(1)BMI=体重(kg)/身長(m)<sup>2</sup>  
(2)検体検出日を日数とした日数

プログラム α

検体番号	検体量 (ng/kg)	年齢 (歳)	BMI <sup>①</sup>	検査名 (PT)	日数 <sup>②</sup>	回収率 (%)	コメント
001	10	15	18.5	下腹部痛	9	測定なし	測定不能ではあるが、検体状況と見られる
				重心	11	測定あり	XXXXXXXXXX検体後に検出しているため、治療に継続あり
002	10	20	18.2	アークリミルトロン スコア増加	125	測定あり	XXXXXXXXXX検体の元の検体もしれないが、治療による影響も否定できないと考え
				倦怠感	126	測定なし	XXXXXXXXXX療法に伴う影響
				血中アルカリホスファターゼ増加	131	測定なし	血中物質の検出に伴う影響
003	20	33	28.4	経尿出血	154	測定なし	検出なしと見られる

(1)BMI=体重(kg)/身長(m)<sup>2</sup>  
(2)検体検出日を日数とした日数




# メリット4:出力のプロセス

Excel

・SASからExcel(DDE経由)に出力して「xxx.xls」を作成

↳ SASプログラム実行中にExcelは操作できない


操作した場合、別シートに結果を出力されるなどの問題が生じる可能性



RTF

・SASから直接「xxx.rtf」を作成

↳ SASプログラム実行中にWordやExcelを操作できる





# デメリット、その他

## デメリット

- ・SASやRTFのバージョンアップによる仕様変更に影響をうける
  - ↳ SAS9.1.3からSAS9.2においても若干見栄えが異なる
- ・フロー図などの自由なデザインの表を作成するのが難しい
  - ↳ 代替えの解析結果のデザインが必要
- ・すべての処理をプログラムで行っているため、些細な修正でもプログラムの再実行が必要
  - ↳ 再度の検証を行う必要性が考えられる
- ・Excelのように印刷縮尺率を変更できないため、1ページに収められる情報量が固定される
  - ↳ 情報量を増やすためにはフォント、余白などの調整が必要  
事前に無理のないmockを計画をしておくことが重要

## その他

- ・Excelについて、Microsoft社からExcel4.0マクロはVBAへの移行が推奨されている



# メリット・デメリットまとめ

	RTF	XLS
編集画面と印刷結果の相違・文字切れ	○：WORDの上では編集画面と印刷結果が同一 ○：標準でセル内改行され全文が出力される	×：完全にはコントロールできない (スタイルの設定などである程度対応可能)
フォント,書式,余白の統一	○：ユーザースタイルで基本部分の統一が容易	×：完全なコントロールは極めて難しい (VBAなどである程度の統一は可能)
出力のプロセス	○：SASからRTFファイルを直接作成するため、 SASプログラム実行中にWordやExcelを使用可能	×：SASプログラム実行中はSASとEXCEL(DDE経由)の 機能を使用するため、Excelは使用できない
出力テンプレート	○：プログラムで完結が可能	×：必要な分だけ事前に出力テンプレートの用意が 必要
毎ページに脚注を表示	○：フッター以外にCOMPUTEステートメントを用いて 毎ページの表の直下に出すことが可能	×：フッターに表示する以外には、 DDE(EXCEL4.0マクロ)での加工が必要
同形式間ファイルの比較	○：Wordの「文書の比較」機能を使用して 比較結果を作成することが可能	×：EXCELには比較機能がないため特別なツールが 必要
グラフ出力	○：SASグラフについては、RTFへ直接出力が可能	△：SASグラフについては、別形式ファイルとして 出力することが必要
文字書式の変更	△：対応可能ではあるが、出力データの加工が必要	○：DDE(EXCEL4.0マクロ)で任意の文字に設定が 可能
セルの結合	△：対応可能ではあるが、出力データの加工が必要	○：DDE(EXCEL4.0マクロ)で任意のセルのマージが 可能
フロー図	×：REPORTプロシジャではフロー図の作成が難しい	○：自由に作成が可能
ページの縮尺の変更	×：ページの縮尺が変更できないため、1ページに 収める情報量が固定される	○：ページの縮尺変更により、1ページに収める 情報量を増やすことが可能
その他	△：全てプログラムで組む必要がある ×：SAS、RTFの仕様変更が大きく影響される	○：DDEを使用する上ではSASの仕様変更による 影響をあまり受けない。 ×：MicrosoftはEXCEL4.0マクロをVBAに移行推奨



# 作成段階



# 使用するデータセット

## データセット

- ・有害事象(3被験者分)の一覧表用SASデータセット
- ・1被験者について複数オブザベーション
- ・\_LINE変数は罫線を引くためのフラグ変数として追加
  - 各被験者の最初のオブザベーション 1
  - 各被験者の最後のオブザベーション 2

使用するデータセット: SAMPLE1										
OBS	被験者番号	投与群 (mg/ke)	年齢 (歳)	BMI	有害事象番号	有害事象名 (PT)	日数	因果関係	コメント	罫線用フラグ
	SUBJID	DOSE	AGE	BMI	AESEQ	AEPT	AESTDY	REL	COMM	_LINE
1	001	10	15	18.5	1	下腹部痛	9	関連なし	原因不明ではあるが、偶発症状と思われる	1
2	001	10	15	18.5	2	悪心	11	関連あり	xxxxxxxxxxx投与後に発現しているため、治験薬に関連あり	2
3	002	10	20	18.2	1	γ-グルタミルトランスフェラーゼ増加	125	関連あるかもしれない	xxxxxxxxxxx投与のための影響かと思われるが、治験薬による影響も否定できないと考える	1
4	002	10	20	18.2	2	倦怠感	126	関連なし	xxxxxxxxxxx療法に伴う影響	.
5	002	10	20	18.2	3	血中アルカリホスファターゼ増加	131	関連なし	抗生物質の使用に伴う影響	2
6	003	20	33	28.4	1	結膜出血	154	関連なし	偶発性と思われる	2





# REPORTプロシジャを使用した SAS-OUTPUTへの出力

# SAS-OUTPUTへの出力

## 出力結果

### 16.2.7 被験者ごとの有害事象一覧

被験者 番号	投与群 (mg/kg)	年齢 (歳)	BMI(1)	有害事象			
				事象名(PT) (MedDRA/J Ver XX.X)	日数(2)	因果関係	コメント
001	10	15	18.5	下腹部痛	9	関連なし	原因不明ではあるが、偶発症状と思われる
				悪心	11	関連あり	xxxxxxxxxxxx投与後に発現しているため、治験薬に関連あり
002	10	20	18.2	γ-グルタミルトラン スフェラーゼ増加	125	関連あるか もしれない	xxxxxxxxxxxx投与のための影響か と思われるが、治験薬による影響 も否定できないと考える
				倦怠感	126	関連なし	xxxxxxxxxxxx療法に伴う影響
				血中アルカリホスファ ターゼ増加	131	関連なし	抗生物質の使用に伴う影響
003	20	33	28.4	結膜出血	154	関連なし	偶発性と思われる

(1: BMI=体重(kg)/(身長(cm)/100)<sup>2</sup>  
(2: 投与開始日を1日目とした日数)

FLOWオプションによる  
折り返し表示



# SAS-OUTPUTへの出力

## 出力プログラム

```
PROC REPORT DATA = SAMPLE1 NOWINDOWS HEADLINE MISSING SPLIT='|';
  COLUMN _LINE
         ('--' SUBJID DOSE AGE BMI ('有害事象' '--' AESEQ AEPT AESTDY REL COMM));

  DEFINE _LINE / SUM NOPRINT;
  DEFINE SUBJID / ORDER ORDER=INTERNAL FLOW WIDTH=8 '被験者|番号';
  DEFINE DOSE / ORDER ORDER=INTERNAL FLOW WIDTH=8 '投与群|(mg/kg)';
  DEFINE AGE / ORDER ORDER=INTERNAL FLOW WIDTH=6 '年齢|(歳)';
  DEFINE BMI / ORDER ORDER=INTERNAL FLOW WIDTH=8 'BMI(1';
  DEFINE AESEQ / ORDER ORDER=INTERNAL NOPRINT;
  DEFINE AEPT / DISPLAY FLOW WIDTH=20 '事象名(PT)|(MedDRA/J Ver XX.X)';
  DEFINE AESTDY / DISPLAY FLOW WIDTH=6 '日数(2';
  DEFINE REL / DISPLAY FLOW WIDTH=10 '因果関係';
  DEFINE COMM / DISPLAY FLOW WIDTH=30 'コメント';

  COMPUTE AFTER AESEQ;
    IF _LINE.SUM = 2 THEN COM = REPEAT("-",110);
    ELSE COM = REPEAT(" ",37) || REPEAT("-",72);
  LINE @3 COM $110.;
  ENDCOMP;

  COMPUTE AFTER;
  LINE @3 '(1:BMI=体重(kg)/(身長(cm)/100)^2';
  LINE @3 '(2:投与開始日を1日目とした日数';
  ENDCOMP;
RUN;
```

・出力する変数の順序

・変数列に対する列幅や  
折り返しなどの設定  
・列幅はバイト単位で設定可能

・罫線行の追加

・脚注の追加

# SAS-OUTPUTへの出力

## 用途

- ・解析結果の検証用資料(目視読み合わせ)として使用  
以前はPRINTプロシジャでの結果を資料として作成

## 特徴

- ・折り返しできるため、解析結果の文字切れを確認できる
- ・PRINTプロシジャではできなかった長い文字列の表示が可能
- ・PRINTプロシジャでの結果より見やすいため、検証作業の効率化

### PRINTプロシジャ

### 検証作業の 効率化

### REPORTプロシジャ

16.2.7 被験者ごとの有害事象一覧

DBS	被験者番号	投与群 (mg/kg)	年齢 (歳)	BMI(1)	事象名(PT) (MedDRA/J Ver XX.X)	日数(2)	因果関係
1	001	10	15	18.5	下腹部痛	9	関連なし
2	001	10	15	18.5	悪心	11	関連あり
3	002	10	20	18.2	γ-グルタミルトランスフェラーゼ増加	125	関連あるかもしれない
4	002	10	20	18.2	倦怠感	126	関連なし
5	002	10	20	18.2	血中アルカリホスファターゼ増加	131	関連なし
6	003	20	33	28.4	結膜出血	154	関連なし

#### DBS コメント

- 原因不明ではあるが、偶発症状と思われる
- xxxxxxxxxxx投与後に発現しているため、試験薬に関連あり
- xxxxxxxxxxx投与のための影響かと思われるが、試験薬による影響も否定できないと考える
- xxxxxxxxxxx療法に伴う影響
- 抗生物質の使用に伴う影響
- 偶発性と思われる

18.2.7 被験者ごとの有害事象一覧

有害事象							
被験者 番号	投与群 (mg/kg)	年齢 (歳)	BMI(1)	事象名(PT) (MedDRA/J Ver XX.X)	日数(2)	因果関係	コメント
001	10	15	18.5	下腹部痛	9	関連なし	原因不明ではあるが、偶発症状と思われる
				悪心	11	関連あり	xxxxxxxxxxx投与後に発現しているため、試験薬に関連あり
002	10	20	18.2	γ-グルタミルトランスフェラーゼ増加	125	関連あるかもしれない	xxxxxxxxxxx投与のための影響かと思われるが、試験薬による影響も否定できないと考える
				倦怠感	126	関連なし	xxxxxxxxxxx療法に伴う影響
				血中アルカリホスファターゼ増加	131	関連なし	抗生物質の使用に伴う影響
003	20	33	28.4	結膜出血	154	関連なし	偶発性と思われる

(1): BMI=体重(kg)/(身長(cm)/100)<sup>2</sup>  
(2): 投与開始日を1日目とした日数



# REPORTプロシジャを使用した RTFへの出力

# RTFへの出力

## 出力結果

### 16.2.7 被験者ごとの有害事象一覧

注釈文字の上付き

日数<sup>(2)</sup>

標準で折り返し表示

被験者 番号	投与群 (mg/kg)	年齢 (歳)	BMI <sup>(1)</sup>	事象名(PT) (MedDRA/J Ver. XX.X)	日数 <sup>(2)</sup>	因果関係	コメント
001	10	15	18.5	下腹部痛	9	関連なし	原因不明ではあるが、偶発症かと思われる
				悪心	11	関連あり	XXXXXXXXXXXX投与後に発現しているため、治験薬に関連あり
002	10	20	18.2	γ-グルタミルトランスフェラーゼ増加	125	関連あるか もしれない	XXXXXXXXXXXX投与のための影響かと思われるが、治験薬による影響も否定できないと考える
				倦怠感	126	関連なし	XXXXXXXXXXXX療法に伴う影響
				血中アルカリホスファターゼ増加	131	関連なし	抗生物質の使用に伴う影響
003	20	33	28.4	結膜出血	154	関連なし	偶発性と思われる

(1: BMI=体重(kg)/(身長(cm)/100)<sup>2</sup>  
(2: 投与開始日を1日目とした日数)

# RTFへの出力

## 出力プログラム (前半)

## ユーザースタイル

```
ODS LISTING CLOSE;  
ODS RTF FILE = "保存パス....rtf" NOTOC_DATA STYLE=USER.STYLE;
```

```
ODS ESCAPECHAR = '~';
```

```
PROC REPORT DATA = SAMPLE1 NOWINDOWS HEADLINE MISSING SPLIT='|'  
  STYLE(Header)={ProtectSpecialChars=OFF Vjust=TOP}  
  STYLE(Lines) = {ProtectSpecialChars=OFF BorderBottomStyle=NONE Just=LEFT};
```

```
COLUMN SUBJID DOSE AGE BMI  
  ("~S={&BRD_BTM.}有害事象" AESEQ AEPT AESTDY REL COMM)  
  _LINE;
```

```
DEFINE _LINE / DISPLAY NOPRINT;  
DEFINE SUBJID / ORDER ORDER=INTERNAL CENTER STYLE={CELLWIDTH=1.5cm} '被験者|番号';  
DEFINE DOSE / ORDER ORDER=INTERNAL CENTER STYLE={CELLWIDTH=1.5cm} '投与群|(mg/kg)';  
DEFINE AGE / ORDER ORDER=INTERNAL CENTER STYLE={CELLWIDTH=1.5cm} '年齢|(歳)';  
DEFINE BMI / ORDER ORDER=INTERNAL CENTER STYLE={CELLWIDTH=1.5cm} 'BMI~{SUPER (1)}';  
DEFINE AESEQ / ORDER ORDER=INTERNAL NOPRINT;  
DEFINE AEPT / DISPLAY LEFT STYLE={CELLWIDTH=3.0cm} '事象名(PT)|(MedDRA/J Ver XX.X)';  
DEFINE AESTDY / DISPLAY CENTER STYLE={CELLWIDTH=1.5cm} '日数~{SUPER (2)}';  
DEFINE REL / DISPLAY CENTER STYLE={CELLWIDTH=1.5cm} '因果関係';  
DEFINE COMM / DISPLAY LEFT STYLE={CELLWIDTH=4.0cm} 'コメント';
```

・SAS-OUTPUTへの出力抑制  
・RTFへの出力開始  
・ユーザースタイルの設定

・部分別のスタイル設定  
STYLE(Header)=: 見出し部分  
STYLE(Lines)=: 脚注部分

・出力する変数の順序  
罫線スタイルのマクロ変数  
&BRD\_BTM.: 下線用設定

・変数列に対する列幅や  
折り返しなどの設定  
・列幅はcm、mm、inで設定可能  
~{SUPER (1)}で上付き

(続き)



# RTFへの出力

## 出力プログラム (後半)

(続き)

```
COMPUTE _LINE;  
  IF _LINE=2 THEN CALL DEFINE("SUBJID", "Style", "Style=[&BRD_BTM.]");  
  IF _LINE=2 THEN CALL DEFINE("DOSE" , "Style", "Style=[&BRD_BTM.]");  
  IF _LINE=2 THEN CALL DEFINE("AGE"   , "Style", "Style=[&BRD_BTM.]");  
  IF _LINE=2 THEN CALL DEFINE("BMI"   , "Style", "Style=[&BRD_BTM.]");  
  CALL DEFINE("AEPT"   , "Style", "Style=[&BRD_BTM.]");  
  CALL DEFINE("AESTDY", "Style", "Style=[&BRD_BTM.]");  
  CALL DEFINE("REL"    , "Style", "Style=[&BRD_BTM.]");  
  CALL DEFINE("COMM"   , "Style", "Style=[&BRD_BTM.]");  
ENDCOMP;  
  
COMPUTE AFTER;  
  LINE '(1:BMI=体重(kg)/(身長(cm)/100)~{SUPER 2}';  
  LINE '(2:投与開始日を1日目とした日数';  
ENDCOMP;  
RUN;  
  
ODS RTF CLOSE;  
ODS LISTING;
```

各セルへの罫線の追加

脚注の追加

RTFへの出力終了

マクロ変数

**BRD\_BTM** = BorderBottomWidth=0.6pt BorderBottomColor=BLACK BorderBottomStyle=SOLID



# 参考までに・・・ユーザースタイルの作成

## ユーザースタイルの プログラム

TEMPLATE プロシジャは設定項目が多いため、  
既存スタイルの継承を利用して  
必要箇所のみ修正を行うのがポイント

```
PROC TEMPLATE;  
  DEFINE STYLE USER.STYLE;  
    PARENT = STYLES.PRINTER;
```

```
  REPLACE FONTS / 'docFont'      = ('M S ゴシック', 8pt)  
                  'headingFont' = ('M S ゴシック', 8pt)  
                  'FixedFont'   = ('M S ゴシック', 8pt)  
                  'BatchFixedFont' = ('M S ゴシック', 8pt)  
                  'FixedStrongFont' = ('M S ゴシック', 8pt, Bold)  
                  'FixedEmphasisFont' = ('M S ゴシック', 8pt, Italic)  
                  'EmphasisFont' = ('M S ゴシック', 8pt, Italic)  
                  'StrongFont'   = ('M S ゴシック', 8pt, Bold)  
                  'TitleFont'   = ('M S ゴシック', 15pt)  
                  'TitleFont2'  = ('M S ゴシック', 15pt);
```

```
  REPLACE COLOR_LIST / 'bg' = WHITE  
                      'fg' = BLACK  
                      'bgH' = WHITE  
                      'link' = BLUE;
```

```
  REPLACE OUTPUT FROM CONTAINER / BORDERWIDTH = 1  
                                  CELLSPACING = 1  
                                  CELLPADDING = 3  
                                  FRAME       = HSIDES  
                                  RULES      = GROUPS;
```

```
  REPLACE TABLE FROM OUTPUT;  
END;  
RUN;
```

・ユーザースタイル「USER.STYLE」を  
作成

・既存スタイルから設定を継承し、  
要部分のみを書き換える

・部分ごとのフォントの設定

・表示色の設定

・セルのマージンや罫線の設定



# まとめ

## Excel との違い

- ・紙に印刷してしまえばほぼ一緒
- ・Wordを使用する上では、編集画面と印刷時の見た目が同一のため、標準では文字切れ、ページ切れしない
- ・事前に無理のないmockを計画しておくことが重要

↳ 文字切れなどを確認する作業がクリアできるため、印刷結果では見えないところに工数軽減のポイント

## SAS-OUTPUT との違い

- ・SAS-OUTPUT、RTFのどちらもプログラムの基本部分はほぼ同じ
- ・ただし、オプションの設定や記述方法が異なる箇所が少ない  
(RTFではデータ部分の強制改行にSPLITオプションが無効、など...)

↳ SAS-OUTPUT用に作成したREPORTプロシジャをRTF出力用に使いまわすのは一筋縄ではいかない可能性もある

## ユーザー スタイル

- ・TEMPLATEプロシジャを用いて作成
- ・基本のフォントや書式を設定して各プログラムに適用
- ・既存スタイル設定を継承して作成することがポイント

↳ 上手く使えば容易に見た目の統一を図ることができる

## SASの バージョン

- ・見た目に関わる罫線、改行、改ページなどの仕様がバージョンごとに異なる

↳ 業務手順書などで使用するSASのバージョンをしっかりと管理



ついでに



# もうちょっと頑張った結果

## 16.2.7 被験者ごとの有害事象一覧

(MedDRA/J Ver XX.X)

被験者 番号	投与群 年齢 BMI <sup>1)</sup>	有害事象			
		事象名 (SOC/PT)	日数 <sup>2)</sup>	因果関係	コメント
001	10(mg/kg) 15(歳) 18.5	胃腸障害	9	関連なし	原因不明ではあるが、偶発症状と思われる
		下腹部痛			
		胃腸障害 悪心	11	関連あり	xxxxxxxxxxxx投与後に発現しているため、治験薬に関連あり
002	10(mg/kg) 20(歳) 18.2	臨床検査	125	関連あるか もしれない	xxxxxxxxxxxx投与のための影響かと思われるが、治験薬による影響も否定できないと考える
		γ-グルタミルトランスフェラーゼ増加			
		全身障害および投与局所様態 倦怠感	126	関連なし	xxxxxxxxxxxx療法に伴う影響
		臨床検査	131	関連なし	抗生物質の使用に伴う影響
003	20(mg/kg) 33(歳) 28.4	血中アルカリホスファターゼ増加			
		眼障害 結膜出血	154	関連なし	偶発性と思われる

(1: BMI=体重(kg)/(身長(cm)/100)<sup>2</sup>  
(2: 投与開始日を1日目とした日数

MeDRAのver表記を見出し上部(右寄せ)に表示

縦セルの結合

条件に合致するセルの書式の変更

SOC/PTを表示してPTの行のみインデント



## 参考資料

---

- ・ Rich Text Format (RTF) Specification Version 1.9.1
- ・ Rich Text Format - Wikipedia  
URL : [http://ja.wikipedia.org/wiki/Rich\\_Text\\_Format](http://ja.wikipedia.org/wiki/Rich_Text_Format)
- ・ SAS TECHNICAL SUPPORT  
URL : <http://www.sas.com/offices/asiapacific/japan/service/technical/faq/index.html>
- ・ SASセミナー ODSによるレポート作成コース ドキュメント



ご静聴ありがとうございました

イーピーエス株式会社

発表者：  
今野 聡

共同発表者：  
富永 一宏  
高橋 伸枝